

## Bayesian Hierarchical Model: Uniform(0,5) prior distribution for the between-tumour standard deviation

```
# CODE ADAPTED FROM: Thall et al (2003)
# Hierarchical Bayesian approaches to phase II trials in diseases with multiple subtypes.
# Statist. Med., 22: 763-780. doi:10.1002/sim.1399
#
# Uniform prior distribution for between-group SD, as recommended by Cunanan et al. (Clinical
Trials, 2019)
#
model{
for (i in 1:numGroups){ # numGroups is k, the number of different probabilities
  x[i] ~ dbin(p[i],n[i]) # In each group, x is the number of responses and n is the number of
patients
  # set up deviance code with correction for zero cells
  x1[i] <- max(x[i],0.1) # zero cell correction
  xhat[i] <- p[i] * n[i] # expected value of the numerators
  xhat1[i] <- max(xhat[i], 0.1) # zero cell correction
  # Deviance contribution with zero cell correction
  dev1[i] <- 2 * (x1[i] * (log(x1[i])-log(xhat1[i]))
    + (n[i]-x1[i]) * (log(n[i]-x1[i]) - log(n[i]-xhat1[i])))
  # deviance contribution for for zero cells
  dev0[i] <- 2 * n[i] * log(n[i]/(n[i]-xhat[i]))
  # deviance contribution
  dev[i] <- dev1[i] * (1>equals(x[i],0)) + dev0[i] * equals(x[i],0)
  # logit model for p
  logit(p[i]) <- rho[i]
  rho[i] ~ dnorm(mu,tau) # RE for log-odds
  # Probability that the response rate for each group is > than targetResp (given as data)
  pg[i] <- step(p[i] - targetResp)
  pg2[i] <- step(p[i] - targetResp2)
}
totresdev <- sum(dev[]) # total residual deviance
# Priors
mu ~ dnorm(mean.Mu, perc.Mu) # pooled mean of log-odds
#tau ~ dgamma(tau.alpha, tau.beta) # used in Thall (2003)
#sd <- 1/sqrt(tau) # between-group sd (log-odds scale)
sd ~ dunif(0,5) # recommended by Cunanan (2019)
tau <- pow(sd,-2)
# predictive distribution
rho.new ~ dnorm(mu,tau) # log-odds response across groups
# convert to probabilities
logit(p.pooled) <- mu # mean probability of response across groups
logit(p.new) <- rho.new # probability response across groups
# predictive probabilities of response rates > targetResp (given as data)
pg.new <- step(p.new - targetResp)
pg2.new <- step(p.new - targetResp2)
}
```

### Data

```
list(x=c(2,2,9,2,6,5,1,6,0,0,0,0), n=c(11,12,7,5,4,4,4,3,2,1,1,1), numGroups=13, mean.Mu=-0.847298,
perc.Mu=0.1, targetResp=0.3, targetResp2=0.1)
```